Quantifier Elimination with *Mathematica* #3

ー東大入試を *2行で* 解くー

2018 年9月16日 Cabri研究会 生越 茂樹

QEのコマンド

f∨g	f || g（or）
f∧g	f && g（and）
¬f	! f（not）
f → g	Implies[f,g]（ =>）
f ⇔ g	Equivalent[f,g]（equiv）
∃x, f	Exists[x,f]
∀x, f	ForAll[x,f]
x∈A	Element[x,A] (elem)
pのQE	Resolve[p,Reals] または Reduce[p,Reals]

- 通常は Resolve の代わりに Reduce でも大丈夫(内部で Resolve をcall）. しかし時間が掛かる.
- ∨, ∧, ⇒ などは“Esc”を使って挟むと入力できる. 例えば “Esc”と”=“と”>”と”Esc” で⇒となる.
- QEはCADを応用している. CAD(Cylindrical Algebraic Decomposition)とは，空間を符号が一定な部分集合に分割すること. なお CADはQE以外にも有効で *Mathematica* のコマンドは「GenericCylindricalDecomposition」と「CylindricalDecomposition」.（後者が数学的な厳密解)
- CADの順序の変更が可能. 例えば「Resolve[p,{y,x},Reals]」とすると x□g(y) の形になる. 「Reduce [p,{y,x},Reals]」より速いことが多い.
- 結果の図示は RegionPlot や RegionPlot3D を使う. 測度は RegionMeasure が大体使える.

§1. 通過領域

[立体射影]

xyz 空間内の原点を中心とする半径 1 の球面
$S=\{(x,y,z)\,|\,x^2+y^2+z^2=1,\ x,y,z$ は実数$\}$
を考え, S 上の定点 $(0,0,1)$ を A とする.
Aとことなる S 上の点 $\mathrm{P}(x,y,z)$ に対し,
直線APと xy 平面の交点を $\mathrm{Q}(u,v,0)$ とする.
点P が $x^2+y^2+z^2=1,\ x\geq\frac{1}{3},\ y\geq\frac{1}{3},\ z\geq\frac{1}{3}$
を満たしながら動くとき，対応する点Q の動く
範囲をuv 平面上に図示せよ．(1993年 東大文 改)



Qは直線AP上より, $\overrightarrow{\mathrm{AP}}=t\,\overrightarrow{\mathrm{AQ}}\Leftrightarrow\begin{pmatrix}x\\y\\z-1\end{pmatrix}=t\begin{pmatrix}u\\v\\-1\end{pmatrix}$

```
ClearAll["Global`*"];
Exists[{x, y, z, t}, x^2 + y^2 + z^2 == 1 && x ≥ 1 / 3 && y ≥ 1 / 3 && z ≥ 1 / 3 && {x, y, z - 1} == t {u, v, -1}];
pred = Reduce[%, {u, v}, Reals] // FullSimplify(* v を u の式で表す*)
```

$$\left(2u==1\ \&\&\ 2v==\sqrt{7}\right)\ ||\ \left(2u==3+\sqrt{7}\ \&\&\ 2v==3+\sqrt{7}\right)\ ||$$
$$\left(\sqrt{-1-(-6+u)\,u}\geq v\ \&\&\ \left(\left(2u>1\ \&\&\ 2u\leq\sqrt{7}\ \&\&\ \sqrt{2-u^2}\leq v\right)\ ||\ \left(2u>\sqrt{7}\ \&\&\ \sqrt{8-u^2}+v\geq 3\ \&\&\ 2u<3+\sqrt{7}\right)\right)\right)$$

```
reg = ImplicitRegion[pred, {u, v}];
RegionPlot[reg]
```



[点射影]

a,b,c を正の実数とする．xyz 空間において，$|x|\le a$，$|y|\le b$，$z=c$ を満たす点(x,y,z) からなる板 R を考える．点光源 P が平面 $z=c+1$ 上の楕円 $\dfrac{x^2}{a^2}+\dfrac{y^2}{b^2}=1$，$z=c+1$ の上を一周する時，光が板 R にさえぎられて xy 平面上にできる影の通過する部分の図をえがき，その面積を求めよ．(1991年 東大理)



a,b,cのままでは図示できないので, a = 2, b = 1, c = 3として解答する.

P(X,Y,c+1), R上の点をR'(x,y,c), 直線PR'とxy平面の交点をQ(u,v,0)とおくと, $\overrightarrow{QR'}=t\,\overrightarrow{QP} \Leftrightarrow \begin{pmatrix} x-u \\ y-v \\ c \end{pmatrix} = t\begin{pmatrix} X-u \\ Y-v \\ c+1 \end{pmatrix}$

```
ClearAll["Global`*"];
a = 2; b = 1; c = 3;
Exists[{x, y, X, Y, t},
  Abs[x] ≤ a && Abs[y] ≤ b && {x - u, y - v, c} == t {X - u, Y - v, c + 1} && X^2 / a^2 + Y^2 / b^2 == 1];
pred = Reduce[%, Reals] (*Resolve では汚い結果になる*)
```

$$\left(-7\le v\le -4 \,\&\&\, -8-2\sqrt{-7-8v-v^2}\le u\le 8+2\sqrt{-7-8v-v^2}\right)\ ||$$
$$(-4<v\le 4 \,\&\&\, -14\le u\le 14)\ ||\ \left(4<v\le 7 \,\&\&\, -8-2\sqrt{-7+8v-v^2}\le u\le 8+2\sqrt{-7+8v-v^2}\right)$$

```
RegionPlot[pred, {u, -15, 15}, {v, -8, 8}, Axes → True, AspectRatio → Automatic]
```



面積は，文字のままで求まる. (14秒ほど掛かる)

```
ClearAll["Global`*"];
Exists[{x, y, X, Y, t}, a > 0 && b > 0 && c > 0 && Abs[x] ≤ a &&
   Abs[y] ≤ b && {x - u, y - v, c} == t {X - u, Y - v, c + 1} && X^2 / a^2 + Y^2 / b^2 == 1];
pred = Resolve[%, Reals]; (*a,b,c のとき Reduce では非常に時間が掛かる*)
{time, area} = RegionMeasure[ImplicitRegion[pred, {u, v}]] // Timing // FullSimplify
```

$$\left\{13.2133,\ \begin{cases} a\,b\,(4+c\,(16+c\,(12+\pi))) & c>0 \,\&\&\, a>0 \,\&\&\, b>0 \\ 0 & \text{True} \end{cases}\right\}$$

§2. 方程式

a,b,c,d を正の数とする．不等式 $\begin{cases} s(1-a)-tb>0 \\ -sc+t(1-d)>0 \end{cases}$ を同時にみたす正の数 s,t があるとき，2次方程式 $x^2-(a+d)x+(ad-bc)=0$ は $-1<x<1$ の範囲に異なる2つの実数解を持つことを示せ (1996年文理)

ForAll, Exists は入れ子にできる.

```
ClearAll["Global`*"];

g[x_] = x^2 - (a + d) x + (a d - b c);
ForAll[{a, b, c, d, s, t},
  a > 0 && b > 0 && c > 0 && d > 0 && s > 0 && t > 0 && s (1 - a) - t b > 0 && -s c + t (1 - d) > 0 ⇒
   Exists[{x1, x2}, g[x1] ≠ g[x2] == 0 && -1 < x1 < x2 < 1]];
Resolve[%, Reals]
```

```
True
```

虚数解を扱う時は Resolve[論理式 , Complexes] と p ∈ Reals , (p|q) ∈ Reals などを組み合わせる.

x についての方程式

$$px^2+(p^2-q)x-(2p-q-1)=0$$

が解を持ち，すべての解の実部が負となるような実数の組(p,q) の範囲を pq 平面上に図示せよ. (1992年東大文)

```
ForAll[x, (p | q) ∈ Reals && p x^2 + (p^2 - q) x - (2 p - q - 1) == 0, Re[x] < 0];
Reduce[%, Complexes] // FullSimplify
RegionPlot[%, {p, -2, 2}, {q, -2, 4}, PlotPoints → 50, Axes → True]
```

$(p == 0 \,\&\&\, -1 < q \leq 0) \;||\; \left(-1+2p < q < p^2 \,\&\&\, (0 < p < 1 \;||\; p > 1)\right)$



(非常に不可解なことに ForAll[x,(p|q)∈Reals&&p x^2+(p^2-q)x-(2p-q-1)==0 ⇒ Re[x]<0] では駄目!)

α,β,γ は x の 3 次方程式;

$$x^3+ax^2+bx+8=0 \quad \cdots①$$

の 3 解とする. (a,b は実数) さらに α,β が

$$\alpha^2+\alpha\beta+\beta^2=\alpha-3\beta-6 \quad \cdots②$$

をみたす時, a と b の値を求めよ. (東大 open 提出)

```
ClearAll[a, b];
(*α が実数の場合と β が実数の場合に分けないと, 時間が非常に掛かる−少なくとも1日以上−*)
Exists[{α, β, γ},
  α^2 + α β + β^2 == α − 3 β − 6 && α + β + γ == −a && α β + β γ + γ α == b && α β γ == −8 && (α | β | a | b) ∈ Reals];
Resolve[%, Complexes]
Exists[{α, β, γ},
  α^2 + α β + β^2 == α − 3 β − 6 && α + β + γ == −a && α β + β γ + γ α == b && α β γ == −8 && (α | a | b) ∈ Reals];
Resolve[%, Complexes]
Exists[{α, β, γ},
  α^2 + α β + β^2 == α − 3 β − 6 && α + β + γ == −a && α β + β γ + γ α == b && α β γ == −8 && (β | a | b) ∈ Reals];
Resolve[%, Complexes]
```

```
False

b == 10 && a == 3

b == −2 && a == −1
```

[注]α, βがともに虚数の場合は false は明らかだが, QEではうまく示せない.

定数 p に対して，3次方程式 $x^3-3x-p=0$ の実数解の中で最大のもの と最小のものの積を $f(p)$ とする．ただし，実数解がただひとつのときには，その２乗を $f(p)$ とする．
(1) p がすべての実数を動くとき，$f(p)$ の最小値を求めよ．
(2) p の関数 $f(p)$ のグラフの概形をえがけ．(1991年理)

```
(*(1)はExists{p,x1,x2}とする. x1 は最小の実数解, x2 は最大の実数解で x1=x2となる事もあり得る*)
ClearAll["Global`*"]; f[x_] = x^3 - 3 x - p;
Exists[{x1, x2, p}, f[x1] == f[x2] == 0 ∧ ForAll[y, y < x1 || y > x2 ⇒ f[y] ≠ 0] ∧ x1 * x2 == m];
Resolve[%, Reals]
```

```
-3 ≤ m ≤ -2 || m > 4
```

```
(*(2)はExists{p,x1,x2}とする. *)
Exists[{x1, x2}, f[x1] == f[x2] == 0 ∧ ForAll[y, y < x1 || y > x2 ⇒ f[y] ≠ 0] ∧ x1 * x2 == m];
Resolve[%, {p, m}, Reals] // FullSimplify;
(*Mathematica10では bugの為に {p,m}が無いとエラーがでる*)
RegionPlot[ImplicitRegion[%, {{p, -6, 6}, {m, -3, 7}}], Axes → True]
```



【別解】解と係数の関係の利用(faster)

```
(*(1)はExists{x1,x2,x3,p}とする. 前半は 実数解を3つ持つ時, 後半は実数解を1つだけ持つ時*)
Exists[{x1, x2, x3, p}, x1 + x2 + x3 == 0 && x1 x2 + x2 x3 + x3 x1 == -3 &&
   x1 x2 x3 == p && (x1 | x2 | p | m) ∈ Reals && x1 ≤ x2 ≤ x3 && x1 * x3 == m] ∨
 Exists[{x1, x2, x3, p}, x1 + x2 + x3 == 0 && x1 x2 + x2 x3 + x3 x1 == -3 &&
   x1 x2 x3 == p && (x1 | p | m) ∈ Reals && ! x2 ∈ Reals && x1 * x1 == m];
Resolve[%, Complexes]
```

```
-3 ≤ m ≤ -2 || m > 4
```

```
(*(2)はExists{x1,x2,x3}とする*)
Exists[{x1, x2, x3}, (x1 | p | m) ∈ Reals,
   x1 + x2 + x3 == 0 && x1 x2 + x2 x3 + x3 x1 == -3 && x1 x2 x3 == p && x1 ≤ x2 ≤ x3 && x1 * x3 == m] ∨
  Exists[{x1, x2, x3}, (x1 | p | m) ∈ Reals && ! x2 ∈ Reals && x1 + x2 + x3 == 0 &&
    x1 x2 + x2 x3 + x3 x1 == -3 && x1 x2 x3 == p && x1 * x1 == m];
Resolve[%, Complexes] // FullSimplify;
RegionPlot[ImplicitRegion[%, {{p, -6, 6}, {m, -3, 7}}], Axes → True]
```



【付録】*Mathematica* は 東京医大に入学できるか？

1 平面上に六点 O，A$_1$，A$_2$，B$_1$，B$_2$，P があり，点 O を始点とする有向線分が表すベクトルの内積の値が

$$\overrightarrow{OA_1}\cdot\overrightarrow{OB_1}=3,\quad \overrightarrow{OA_1}\cdot\overrightarrow{OB_2}=4,\quad \overrightarrow{OA_1}\cdot\overrightarrow{OP}=14,$$
$$\overrightarrow{OA_2}\cdot\overrightarrow{OB_1}=5,\quad \overrightarrow{OA_2}\cdot\overrightarrow{OB_2}=7,\quad \overrightarrow{OA_2}\cdot\overrightarrow{OP}=23,$$
$$\overrightarrow{OP}\cdot\overrightarrow{OB_1}=22,\quad \overrightarrow{OP}\cdot\overrightarrow{OB_2}=35$$

であった．このとき，$\overrightarrow{A_1A_2}\cdot\overrightarrow{PB_1}=\boxed{\text{アイ}}$ であり，線分 OP の長さは $\mathrm{OP}=\sqrt{\boxed{\text{ウエ}}}$ である．

```
ClearAll["Global`*"]
A1 = {a, 0}; A2 = {a3, a4}; B1 = {b1, b2}; B2 = {b3, b4}; p = {x, y};
cons = A1.B1 == 3 && A1.B2 == 4 && A2.B1 == 5 && A2.B2 == 7 && A1.p == 14 && A2.p == 23 && B1.p == 22 && B2.p == 35;

Exists[{a, a3, a4, b1, b2, b3, b4, x, y}, cons && (A2 - A1).(B1 - p) == k1];
Resolve[%, Reals]
Exists[{a, a3, a4, b1, b2, b3, b4, x, y}, cons && x^2 + y^2 == k2];
Resolve[%, Reals]
```

```
k1 == -7

k2 == 97
```

2 (1) a，b，c を定数とする．関数 $f(x)=x^5+ax^4+bx^3+cx^2$ が，$x=0$ において極大値 0 をとり，$x=5$ において極小値 0 をとるときについて考える．
このとき，定数 b の値の範囲は $\boxed{\text{アイ}}<b<\boxed{\text{ウエ}}$ である．
また，このときの $f(x)$ は $x=0$ 以外でも極大値をとり，その極大値 M の範囲は $\boxed{\text{オ}}<M<\boxed{\text{カキク}}$ である．

(2) 関数 $f(x)=(ax)^{-\frac{(bx)^{cx}}{3}}$ の $x=1$ における微分係数は $a=8$，$b=e^{-1}$，$c=-\log 2$ であるとき $f'(1)=\dfrac{\boxed{\text{ケコ}}}{\boxed{\text{サ}}}$ である．ただし，対数は e を底とする自然対数とする．

```
(*1*)
f[x_] := x^5 + a x^4 + b x^3 + c x^2;
cons = f'[0] == 0 ∧ f[0] == 0 ∧ f''[0] < 0 ∧ f'[5] == 0 ∧ f[5] == 0 ∧ f''[5] > 0;
Exists[{a, c}, cons];
Resolve[%, Reals] // Simplify
Exists[{a, b, c, p}, cons && f'[p] == 0 && f[p] == M && f''[p] < 0 && p ≠ 0];
Resolve[%, Reals] // Simplify
```

```
25 < b < 75

0 < M < 108
```

```
(*2*) ClearAll["Global`*"]
f[x_] = ((a x)^(-(b x)^(c x) / 3)) /. {a → 8, b → 1 / E, c → -Log[2]};
f'[1]
```

$-\dfrac{1}{6}$

3 条件

$$a_1 = 1, \quad a_{n+1} = a_n \cos \frac{n^2\pi}{3}$$

によって定められる数列 $\{a_n\}$ について考える.

(1) $a_6 = \dfrac{\boxed{アイ}}{\boxed{ウエ}}$ であり，$\displaystyle\sum_{n=1}^{6} a_n = \dfrac{\boxed{オカ}}{\boxed{キク}}$ である.

(2) $\log_2 |a_{70}| = \boxed{ケコサ}$ であり，$\displaystyle\sum_{k=1}^{\infty} a_{6k+1} = \dfrac{\boxed{シス}}{\boxed{セソ}}$ である.

(3) $\displaystyle\sum_{n=1}^{\infty} a_n = \dfrac{\boxed{タチ}}{\boxed{ツテ}}$ である.

```
ClearAll[a]

sol = RSolve[{a[n + 1] == a[n] Cos[n^2 Pi / 3], a[1] == 1}, a[n], n];
a[n_] = (a[n] /. sol)[[1]] // FullSimplify
```

$$(-1)^{1+\text{Floor}\left[\frac{1}{6}(-5+n)\right]+\text{Floor}\left[\frac{1}{6}(-4+n)\right]+\text{Floor}\left[\frac{1}{6}(-3+n)\right]}\ 2^{-3+\text{Ceiling}\left[\frac{2-n}{6}\right]+\text{Ceiling}\left[\frac{3-n}{6}\right]+\text{Ceiling}\left[\frac{5-n}{6}\right]-\text{Floor}\left[\frac{n}{6}\right]}$$

```
(*1*)
a[6]
Sum[a[k], {k, 1, 6}]
```

$-\frac{1}{16}$

$\frac{21}{16}$

```
(*2*)
Log[2, Abs[a[70]]]
Sum[a[6 k + 1], {k, 1, Infinity}]
```

-46

$-\frac{1}{17}$

```
(*3*)
Sum[Sum[a[6 k + r], {k, 0, Infinity}], {r, 1, 6}]
```

$\frac{21}{17}$

4 関数 $f(x)$ を

$$f(x)=\frac{1}{2}x+\frac{1}{8}x^3 \quad (x \geqq 0)$$

により定め，座標平面上の曲線 C を $C: y=f(x)$ とし，曲線 C と y 軸と直線 $y=2$ で囲まれた図形 F について考える.

(1) 図形 F の面積は $\dfrac{\boxed{\text{ア}}}{\boxed{\text{イ}}}$ である.

(2) 関数 $g(x)$ を $g(x)=f^{-1}(x)^2$ により定義する．このとき $g^{-1}(2)=\dfrac{\boxed{\text{ウ}}\sqrt{\boxed{\text{エ}}}}{\boxed{\text{オ}}}$ である.

ただし，$f^{-1}(x)$ と $g^{-1}(x)$ はそれぞれ $f(x)$ と $g(x)$ の逆関数とする.

(3) 図形 F を y 軸のまわりに回転してできる立体の体積は $\dfrac{\boxed{\text{カキ}}}{\boxed{\text{クケ}}}\pi$ である.

```
ClearAll[f, x]
f[x_] = 1 / 2 x + 1 / 8 x^3;
F = ImplicitRegion[f[x] < y < 2 && x > 0, {x, y}];
Show[Plot[{f[x], 2}, {x, -1, 3}], RegionPlot[F]]
```




```
(*1*)
RegionMeasure[F]
```

$\frac{5}{2}$

(*2*) (InverseFunction は逆関数の枝のうち適当にひとつを選ぶので,出題者の意図した解答と符号が異なる)

```
f[x_] = 1 / 2 x + 1 / 8 x^3;
h[x_] = x^2;
g = Composition[h, InverseFunction[f]]; (*g=(f^-1)^2*)
InverseFunction[g][2]
```

$-\frac{3}{2\sqrt{2}}$

上では $h^{-1}[x] = -\sqrt{x}$ の枝を取っているので符合が異なる! そこで手動で解答する

```
Composition[f, Sqrt[#] &][2]
```

$\frac{3}{2\sqrt{2}}$

```
(*3*)
(*バウムクーヘン分割の利用*)
Integrate[2 Pi x (2 - f[x]), {x, 0, 2}]
```

$\frac{56\pi}{15}$

```
(*RegionMeasure の利用. 答えは一致するが, ちょっと時間が掛かる*)

ImplicitRegion[2 > z > f[Sqrt[x^2 + y^2]], {x, y, z}];
RegionMeasure[%]
```

$\frac{56\pi}{15}$

5 座標平面上の曲線 $(x^2+y^2)^2=x^3-3xy^2$ を描け．
5 の解答は，数学解答用紙(省略)に解答の曲線だけを記入せよ．解答の曲線以外の補助線や目盛りなどの数値を新たに記入してはならない．

```
ContourPlot[(x^2 + y^2)^2 == x^3 - 3 x y^2, {x, -0.7, 1.1}, {y, -1, 1}, PlotPoints → 50, Axes → True]
```